
Course Plan
Session I:  Background

• Rewriting Logic
• Maude

Session II:  Simple Signaling Processes
• Biological processes -- informally
• Pathway Logic representation of signaling

•SmallKB - a small network

Session III:  Interactive Analysis
• Transformation to Petri nets
• The Pathway Logic Assistant

Session IV: Using the Pathway Logic Assistant
• Demo -- SmallKB live
• The Egf Stimulation Model

• Curation
• Demo



Lecture I: Outline

Logic
•  Propositional Reasoning
•  Equational Reasoning
•  ADTs
•  Rewriting Logic

Maude <http://maude.cs.uiuc.edu/>
•  Functional modules (Equational)
•  System modules
•  Search and model-checking
•  Reflection



About Logic

Making description and reasoning precise

• Language
• for describing things and/or properties
• given by a signature and rules for generating
additional elements (terms, formulas)

• Models--for interpreting to language elements
• a mathematical structure
• interpretation of terms
• satisfaction of formulas: M |= wff

• Reasoning
• rules for inferring what !holds in all models



Propositional/Boolean Logic

• Language
• Signature -- a set of propositions  P,Q, ...
• Formulas built from propositions using boolean
constants and connectives (no terms)
• true, false,  P and Q, not P, ...

• Model-- A valuation of propositions
• V : Prop -> {T,F}
• Extended to formulas:

• V(P and Q) = if  (V(P) = T and V(Q) = T) then T else F
• V |= wff if V(wff) = T
•

• Reasoning
• simplest is truth table



Propositional/Boolean Logic II

• Basic truth table

•
•Defined operators
• A or B := not( not(A) and not(B))
• A impl!! B := not(A) or B

FTFF

FTTF

FFFT

TFTT

A and Bnot(A)BA



Propositional/Boolean Logic: Example

• V: A,B,C -> T,T,F
• V( !A and B and not(C)) = T

• otherwise
• V( A and B and not(C)) = F

T

F

T

F

T

F

T

F

not(C)

F

F

F

F

F

F

T

T

(A and B)

FTTF

FFTF

FTFT

FFFT

F

F

T

F

(A and B)  and not(C)

F

T

F

T

C

FF

FF

TT

TT

BA

Note:
!   A and B and not(C)
   is equivalent to
   not(A and B impl C)

Proof:  Compute truth table from the basic table



Example as a Boolean Circuit



Equational Logic: Language

•Signature
• a set of operation symbols with arity, o : On
• a set of variable symbols, X

• Terms   T
• x : X
• c : O0
• f(t1,...,tn)  for f : On  and t1,...,tn : T

•Formulas --  t0 = t1



Equational Logic: Models

• M = < D, I >
• D a set (of data)
• I : O0  -> D,  I : On -> (Dn -> D)
• V : X -> D variable assignment

• Interpretation relative to a variable assignment
•I [V] (x) = V(x)
• I [V] (f(t1, ..., tn)) = I(f)(I[V](t1), ..., I[V](tn))

• Satisfaction
• M |= t0 = t1 [V]  if I[V] (t0) = I[V] (t1)



Equational Logic: Reasoning Rules

• equivalence
• t = t
• t0 = t1  |--  t1  = t0
• t0 = t1 , t1  = t2 |-- t0  = t2

• congruence
• t1 = t1’  ... tn  = tn’ |-- f(t1, ..., tn) = f(t1

’, ..., tn’)

• substitution
• t = t’ |-- t[S] = t’[S]



Equational Theory

• ET = Equational Signature + Equational Axioms (E)

• M |= ET if M |= eq for eq : E

• t[E]  -- the equivalence class of t wrt E

• Initial model  < T/E,  I/E>
• no junk  -- every element is denotation of a term
• no confusion -- terms have equal denotations iff
they are in the same equivalence class
• proof by induction on term structure



Equational Theory: Peano Numbers
•  Signature

• Ops:  0 : O0 , s : O1,  +,m : O2
•  Vars: x, y, ... : X

•  Equations:
1.  x + 0 = 0
2.  x + s y = s(x + y)
3.  0 m x = 0
4.  s(x) m 0 = s(x)
5.  s(x) m s(y) = x m y

•  Calculating
 s s s 0 + s s 0

= s(s s s 0 + s 0)   --- 2[x := s s s 0, y:= s 0]
= s s (s s s 0 + 0)  --- 2[x := s s s 0, y:=  0]
= s s s s s 0          --- 1 congruence



ADTs (Abstract Data Types)

An ADT consists of one or more sets of values, together
with a set of operations

  (D1 ... Dk,  c1 ... cm, f1 ... fn)
An initial model of an equational theory
0 Values are given by constructors: c(t1,...,tl)
0 Functions f1 .. fn are defined by mathematical

relations (usually equations).



Equational Logic: Variations

• Sorted EqL -- Signature includes sorts (data types)
and operation arity is sorted ( + : Nat Nat -> Nat)

• Order Sorted EqL -- SortedEqL with a partial order
on the sorts  (Zero < Nat, NZNat < Nat)

• Membership EqL -- connected components of the
sort partial order become Kinds, which can contain
additional `error’ elements.   Terms are typed at the
kind level.



Rewriting Logic

Q:   What is rewriting logic?

A1:  A logic for executable specification and
analysis of software systems, that may be
concurrent, distributed, or even mobile.

A2: A logic to specify other logics or languages

A3: An extension of equational logic with local
rewrite rules to express
        1.  concurrent change over time
        2.  inference rules



Rewrite Theories

0 Rewrite theory:  (Signature, Labels, Rules)

0 Signature:  (Sorts, Ops, Eqns) -- an equational theory
– Describe data types,  system state

0 Rules have the form   label : t => t’  if cond

0 Rewriting operates modulo equations
– Describes computations or deductions, also modulo equations



Deduction/Computation Rules

reflexivity: replacement:
congruence:

f f

one step rewrite:

closed under



Rewriting Logic is Reflective

A reflective logic is a logic in which important aspects
of its metatheory (entailment relation, theories,
proofs) can be represented at the object level in a
consistent way.

This has many applications:
0 Transforming, combining rewrite theories
0 Execution / proof strategies
0 Meta tools:  theorem provers, coherence checkers ...
0 Language extensions: object-oriented, real-time, ...
0 Higher-order capabilities in a first-order framework
0 Model of reflection for concurrent objects
0 Domain specific assistants



Rewriting Logic as a
Semantic Framework

A wide variety of models of computation can be
naturally expressed as rewrite theories

0 Lambda Calculus, Turing Machines
0 Concurrent Objects and Actors
0 CCS, CSP,  pi-calculus, Petri Nets
0 Chemical Abstract Machine, Unity
0 Graph rewriting, Dataflow, Neural Networks
0 Real-time Systems
0 Physical Systems (Biological processes)



Rewriting Logic as a
Logical Framework

A wide variety of logics have a natural representation
as rewrite theories

0 Equational logic
0 Horn logic
0 Linear logic
0 Quantifiers
0 Higher-order logics (HOL, NuPrl)
0 Open Calculus of Constructions
0 Rewriting logic



What Rewriting ! Logic Is/IsNot

0 A rewrite theory plus a term describes a state
transition system

– states can have rich algebraic structure
– transitions are local and possibly concurrent

0 The equational part of a rewrite theory is similar to a
term rewriting system (modulo ACI axioms)

– it is usually desirable for equations to be CR and terminating
– rewrite rules are often non-deterministic and non-terminating



Maude

0 Maude is a language and environment ! based on
rewriting logic

0 See:   http://maude.cs.uiuc.edu
0 Features:

– Executability -- position /rule/object fair rewriting
– High performance engine --- {ACI} matching
– Modularity and parameterization
– Builtins --  booleans, number hierarchy, strings
– Reflection -- using descent and ascent functions
– Search and model-checking



Impact

rapid 
prototyping

state space 
search

S |=Φ
model 

checking

Maude Formal Methodology

S |- Φ
theorem 
provng

model building



ADTs in Maude

ADTs are specified in Maude using functional modules
      fmod <modname> is
        <imports>      *** reuse, modularity
        <sorts>        *** data types and subtyping
        <opdecls>      *** names/and arities of operations
        <eqns>         *** how to compute functions
      endfm

0 The module components can appear interleaved and in
any order

0 The semantics of an fmod is an initial model
– no junk---every data value is constructable
– no confusion---two terms are equal only forced by the

equations



Maude System Specifications

A Maude specification has two parts
0 An equational part describing structure and

properties of system states (and ADT)
0 A rules part specifying how the system might change

over time.



Operator declarations

The <opdecls> part of a module consists of operator declarations
having  the form

  op <opname> : <argSorts> -> <resultSort> [<attributes>] .

opname can be an identifier (without special characters)  or
identifiers interleave with `_’s  (mixfix notation).

<attributes> for binary operators include assoc, comm, id:
<term>

Including assoc declares the operator to b associative, ...

*** unary constructor
  op s_! : Nat !!!!-> Nat [ctor] .
*** unary operator
  op sub1! : Nat !!!!-> Nat .
***  associative commutative infix operator with identity
  op _+_! : Nat Nat -> Nat [assoc comm id: 0] .



The NAT and NATLIST Data Types

The natural numbers are specified by
      fmod NAT is
        sorts Zero NzNat Nat .
        subsort Zero NzNat < Nat .
        op 0 : -> Zero [ctor] .
        op s_ : Nat -> NzNat [ctor] .
            ....
      endfm

  Examples: 0, s s s 0 (also printed as 3)

Lists of natural numbers are specified by
     fmod NATLIST is

    pr NAT .
    sort NatList .  subsort Nat < NatList .
    op nil : -> NatList .
    op _ _ : NatList NatList -> NatList [assoc id: nil] .
        ....
  endfm

Example lists:  nil,  1 2 3



Defining NatList Functions I

Definition by pattern matching (extending NATLIST)

  var n: Nat. var nl: NatList .
  op head : NatList ~> Nat .
  op tail : NatList -> NatList .
  eq head(n nl) = n .
  eq tail(n nl) = nl .   eq tail(nil) = nil .

Consider the list !: ! 1 2 3
It matches  the pattern (n nl) with n := 1, nl := 2 3

Maude> reduce head(1 2 3) .    Maude> red tail(1 2 3) .
result NzNat: 1                result NatList: 2 3

Maude> red head(nil) .         Maude> red tail(nil) .
result [Nat]: head(nil)        result NatList: nil



Defining NatList Functions II
Definition by recursion

Define a function `sum’ that adds the elements of a
NatList.

  var n : Nat. var nl : NatList .
  op sum : NatList -> Nat .
  eq sum(nil) = 0 .
  eq sum(n nl) = n + sum(nl) .

The equations are used to reduce terms to canonical form
  sum(1 2 3) = 1 + sum(2 3) = 1 + 2 + sum(3) = 1 + 2 + 3 = 6

  Maude> reduce  sum(1 2 3) .
  result NzNat: 6



Defining NatList Functions III
Conditional equations

Define a function `isElt’ that takes a Nat and a NatList
and returns true if the Nat is in the NatList and false
otherwise.

  vars n n’: Nat.   var nl: NatList .
  op isElt : Nat NatList -> Bool .
  eq isElt(n, n’ nl) =
            if n == n’ then true else isElt(n,nl) fi .
  eq isElt(n, nil) = false .

  Examples:
    isElt(4, 4 6 3) = true
    isElt(4, 7 2 4 6) = isElt(4,2 4 6)
    isElt(4,nil) = false



Defining NatList Functions IV

Alternate definition of `isElt’: via AC pattern matching

  vars n : Nat .   vars nl nl’: NatList .
  op isElt : Nat NatList -> Bool .
  eq isElt(n, nl n nl’) = true .
  eq isElt(n, nl) = false [owise] .

Examples:
isElt(4, 7 2 4 6 3) matches the 1st equation
 taking    n := 4, nl := 7 2, nl’ := 6 3
  and so reduces to true
isElt(4, 7 2 3) does not match the 1st equation, so

reduces to false by the [owise] equation



Using Maude

maude is an alias to the ! maude executable (maude.linux)
To try the NatList examples:
0 $> cd <pathToExamples>MaudeExamples
0 $> maude nat-list

                 \||||||||||||||||||/
         --- Welcome to Maude --
           /||||||||||||||||||\
    Maude alpha89b built: Apr 25 2007 16:19:05
    Copyright 1997-2007 SRI International
          Thu Jun 28 07:42:00 2007
0 Maude> red head(1 2 3) .
  reduce in NATLIST : head(1 2 3) .
  rewrites: 1 in 0ms cpu (0ms real) (~ rewrites/second)
  result NzNat: 1



Caveats

Caveats for writing Maude specifications:
0 Don’t forget the period at the end of a declaration or

statement!  .
0 Check keywords / symbols are correct.

– correct use of fmod, mod
– op vs ops
– variables not used before bound

0 Make sure all cases are covered in defining equations.



Specifying behavior/dynamics

System dynamics are specified in system modules using
rewrite rules

    mod <modname> is

      *** functional part
        <imports>      *** reuse, modularity
        <sorts>        *** data types and subtyping
        <opdecls>      *** names/and arities of operations
        <eqns>         *** how to compute functions
      ***
        <rules>
      endm

A system module defines a set of computations  (aka
derivations) over the ADT specified by the functional
part.



Rule declarations

The <rules> part of a system module consists of rule declarations
having one of the forms

  rl[<id>]: <lhs> => <rhs>  .

 crl[<id>]: <lhs> => <rhs> if <cond> .

<lhs>, <rhs>, <cond>   are terms, possibly containing variables.
A rule applies to a term  T if there is a substitution S (mapping

variables to terms)  such that  S<lhs> is a subterm of T  (<lhs>
matches a subterm of T) and S <cond> rewrites to true.

In this case T can be rewritten by replacing the matched subterm by
the matching instance of <rhs> (S <rhs>).



Rule application

Suppose we have the following rules (in a suitable module).

  rl[fa2b]: f(a,x) => f(b,x) .

  rl[hh2h]: h h(y) => h(y) .

then
  g(c,f(a,d)) => g(c,f(b,d))

using [fa2b] and congruence

and
  h h(g(c,f(a,d))) => h(g(c,f(b,d)))

also using replacement.

Before applying a rewrite rule, Maude reduces a term to canonical
form using equations.  The rule lhs is not reduced.



Petri Nets

Petri nets (Place-transition nets) are a formalism for
modeling concurrent system behavior.

0 Places represent properties or component features.

0 System state is represented by marking of places
(what properties hold or features are present)

0 Transitions specify state change as transfer of marks
from one place to another

0 Petri nets have a natural graphical representation  and
a natural representation in Maude



A simple Petri net computation



Petri Net Example
A vending machine

Buy-c Buy-a change

c a q

$

4



The vending machine in Maude

mod VENDING-MACHINE is
  sorts Coin Item Place Marking .
  subsorts Coin Item < Place < Marking .
  op null : -> Marking .           *** empty marking
  ops $ q : -> Coin .
  ops a c : -> Item .
  op _ _ : Marking Marking -> Marking
            [assoc comm id: null] .  *** multiset
  rl[buy-c]: $ => c .
  rl[buy-a]: $ => a q .
  rl[change]: q q q q => $ .
endm



What about AC Rewriting?

q M:Marking q matches  a subterm of a q c $ q q
 with M:Marking := c $
q M:Marking a M1:Marking matches  a subterm of
$ q c $ q a

 with M:Marking := c $ q and M1:Marking := nil

For terms constructed from operators that are
declared with attributes such as assoc, comm and / or
id, rule matching is extended to match AC terms to
segments.



Using the vending machine

Maude> rew $ $ $ .
result Marking: q a c c

Maude> search $ $ $ =>! a a M:Marking .

Solution 1 (state 8)
M:Marking --> q q c

Solution 2 (state 9)
M:Marking --> q q q a

No more solutions.
states: 10  rewrites: 12)

What is one way to use 3 $s?

How can I get 2 apples with 3 $s?



Model checking

Algorithm for determining if M |= P  (M satisfies P)
where M is a `model’ and `P’ is a property.

In our case a model is a Maude specification of a
system together with a state of interest.

A property is a `temporal logic’ formula to be
interpreted as a property of computations of the
system (linear sequences of states generated by
application of rewrite rules).



Model checking

Temporal formulas are built from propositions about
states, using boolean connectives, and temporal
operators [] (always) and <> (eventually).

  Satisfaction for M |= A  is axiomatized by equations

   M |=  P  if   C |= P for every computation  C of M

   C |= A if the first state of C satisfies A

   C |= [] P if  every suffix of C satisfies P

  C |= <>P if  some suffix of C satisfies P



Model checking: Defining properties

mod MC-VENDING-MACHINE is
  inc VENDING-MACHINE .  inc MODEL-CHECKER .  inc NAT .

  op vm : Marking -> State [ctor] .
  op countPlace : Marking Place -> Nat .
  op value : Marking -> Nat .  *** in units of quarters
  ....

  ops fiveQ lte4Q : -> Prop .
  ops nCakes nApples val : Nat -> Prop .

  eq vm(M) |= fiveQ = countPlace(M,q) == 5 .
  eq vm(M) |= lte4Q = countPlace(M,q) <= 4 .
  eq vm(M) |= nApples(n) = countPlace(M,a) == n .
  eq vm(M) |= val(n) = value(M) == n .
endm



Model checking the vending machine I

Starting with 5 $s,  can we get 6 apples without accumulating more
than 4 quarters?  Model check the claim that we can't.

Maude>
   red modelCheck(vm($ $ $ $ $),[]~(lte4Q U nApples(6))) .
result ModelCheckResult: counterexample(
  {vm($ $ $ $ $),'buy-a}
  {vm($ $ $ $ q a),'buy-a}
  {vm($ $ $ q q a a),'buy-a}
  {vm($ $ q q q a a a),'buy-a}
  {vm($ q q q q a a a a),'change}
  {vm($ $ a a a a),'buy-a}
  {vm($ q a a a a a), 'buy-a},
  {vm(q q a a a a a a),deadlock})

A counterexample to a formula is a pair of transition lists
representing an infinite compuation which fails to satisfy the
formula.  A transition is a state and a rule identifier.  The second
list (deadlock) represents a loop.



Model checking the vending machine II

Starting with 5 $s,  can we get 5 quarters then 6 apples?

Maude>
red modelCheck(vm($ $ $ $ $),
               []~(<>fiveQ /\ (fiveQ |-> nApples(6))) .
result ModelCheckResult: counterexample(...)

Is value conserved?
Maude> red modelCheck(vm($ $ $ $ $),[]val(20) .
result Bool: true



Extracting Paths from CounterExamples
A simple path from a given initial state S0, to a state
satisfying a property P is a list of rules together
with a state S1 satisfying P such that applying the
rules starting with S0 leads to S1.

One way to find a simple path is to model check the
assertion that from S0 no state can be reached
satisfying P:  modelCheck(S0, ~ <> P).    If there is a
reachable state satisfying P, a counterexample will
be returned.   The counterexample contains the list
of rules applied.



Using findPath

The module SIMPLE defines a function findPath that
extracts a simple path from a counterexample and a
property.

mod FP-VENDING-MACHINE is
  inc MC-VENDING-MACHINE .
  inc SIMPLE .
endm

Maude> red findPath(vm($ $ $ $), nApples(5)) .
result SimplePath:
  spath(‘buy-a ‘buy-a ‘buy-a ‘buy-a ‘change ‘buy-a,
         vm(q a a a a a)) .



A Vending Machine Mall

To illustrate systems states with more structure,
consider a `Mall’ which contains possibly multiple
vending machines, each vending a specific item.
In addition the Mall contains `Goods’,  items
purchased, and coins available to make purchases.

We add rules to put coins into machines and extract
items purchased.   The orginial vending machine rules
are used to turn coins into items.



The Mall specification

mod VENDING-MALL is
  inc VENDING-MACHINE .
  sorts Machine Goods Mall .
  subsort Machine Goods < Mall .

  op mt : -> Mall .
  op __ : Mall Mall -> Mall [ctor assoc comm id: mt] .

  op m : Item Marking -> Machine [ctor] .
  op g : Place -> Goods [ctor] .

  var M : Marking . var C : Coin . var I : Item .

   rl [coin-in] : m(I, M) g(C) => m(I, M C) .
  rl [apple-out] : m(a, M a) => m(a, M)  g(a) .
  rl [cake-out] : m(c, M c) => m(c, M)  g(c) .

endm



Shopping in the Mall

Consider an initial state with an two vending machines, one for
apples, one for cakes, and 4 $s.

Default shopping yields 2 apples.

Maude> rew m(a,null) m(c,null) g($) g($) g($) g($) .
result Mall: g(a) g(a) m(a, q q c c) m(c, null)

Can I get  4 apples and a cake?

Maude> search m(a,null) m(c,null) g($) g($) g($) g($) =>!
X:Mall g(a) g(a) g(a) g(a) g(c) .
No solution.

But I can get 4 apples and 4 quarters stuck in the machine!

search m(a,null) m(c,null) g($) g($) g($) g($)
        =>!  m(a, q q q q) X:Mall .
Solution 1 (state 725)
X:Mall --> g(a) g(a) g(a) g(a) m(c, null)



Reflection: Meta-representation of terms

• A constant is meta-represented by a quoted identifier
 (qid) where the identifier is its name dotted with the name
of its (least) sort.

• ‘$.Coin meta-represents the term $
• ‘a.Item meta-represents the term a

• A variable is meta-represented by a qid containing its
name and the name of its  sort separated by :

• ‘M:Marking meta-represents the variable M:Marking
•The meta-representation of a composite term has the
form op[t1,...,tk]  where op is a qid nameing the operator

• '__['q.Coin,'q.Coin,'q.Coin,'q.Coin] is the meta-
representation of q q q q .



Reflection: Ascent/Descent functions

•upTerm(a) = ‘a.Item the meta-representation of a term

•upRls(modname,bool)
• the set of rules (meta-represented) in the module with name
 modname (and its imports if bool is true)

• metaReduce(module, term) = {term’,type} = res

• metaXapply(module,term,rid,subst,start,bound,n)
   = res = {term’,type,subst’,context}

• start,bound bound the search for a match
• n is the solution number

•getTerm(res) = term’



Reflection example: Module analysis

fmod CONSUMERS is
  inc MY-META .
  var M : Module .   var T : Term .
  var R : Rule .   var RS : RuleSet .

  op consumes : Module Rule Term -> Bool .
  eq consumes(M,R,T) =           *** assumes M defines ‘has
      getTerm(metaReduce(M,'has[getLhs(R),T] ))
        == 'true.Bool .

  op consumerRules : Module Term -> QidSet .
  op consumerRules : Module RuleSet Term -> QidSet .

  eq consumerRules(M,T) =
       consumerRules(M,upRls(getName(M),true),T) .
  eq consumerRules(M,none,T) = none .
  eq consumerRules(M,R RS,T) =
   (if consumes(M,R,T) then getRuleId(R) else none fi);
       consumerRules(M,RS,T) .
endfm



Reflection Example: Module analysis cntd.

mod VEND-X is
  inc VENDING-MACHINE .

  vars M0 M1 : Marking .
  op has : Marking Marking -> Bool .

  eq has(M0 M1, M1) = true .
  eq has(M0, M1) = false [owise] .
endm

select CONSUMERS .

Maude>  red consumerRules(['VEND-X],'$.Coin) .
result: QidSet 'buy-a ; buy-c

Maude>  red consumerRules(['VEND-X],'q.Coin) .
result: Qid 'change



Reflection example: Strategy

fmod METAREWRITE-LIST is
  inc MY-META .
  var M : Module .   vars T T’: Term .
  var res : Result4Tuple? .
  var rid : Qid .   var ql : QidList .

  op metaRewList : Module QidList Term -> Term .
  eq metaRewList(M,nil,T) = T .
  ceq metaRewList(M,rid ql,T) = metaRewList(M,ql,T')
    if res := metaXapply(M,T,rid,none,0,unbounded,0)
    /\ T' := if res :: Result4Tuple
             then getTerm(res)
             else T fi .
endfm



Reflection example: Strategy cntd.

Maude> red metaRewList(['VENDING-MACHINE],
            'change 'buy-a,
            '__['q.Coin,'q.Coin,'q.Coin,'q.Coin]) .

result GroundTerm: '__['q.Coin,'a.Item]

Maude> red metaRewList(['VENDING-MACHINE],
             'buy-a 'change,
             '__['q.Coin,'q.Coin,'q.Coin,'q.Coin]) .

result Constant: '$.Coin



Maude Exercises

The folder MaudeExamples contains the code for all
the examples.

See README.txt for a summary of the file contents

See exercises.txt for instructions


